```
> restart
```

Maple worksheet with the examples of the paper: Bertrand Teguia Tabuguia, Arithmetic of D-algebraic Functions

May 2023

Prerequisite: Make sure the NLDE package can be used as a library with your Maple version.

Example 4.

```
> ADE1:=diff(y[1](x),x)^2+y[1](x)^2-1=0
```

$$ADE1 := \left(\frac{d}{dx} y_1(x)\right)^2 + y_1(x)^2 - 1 = 0 \quad (1.1)$$

```
> simplify(eval(ADE1,y[1](x)=lambda[1]*cos(x)+lambda[2]*sin(x)))
```

$$\lambda_1^2 + \lambda_2^2 - 1 = 0 \quad (1.2)$$

```
> ADE2:=diff(y[2](x),x)=y[2](x)
```

$$ADE2 := \frac{d}{dx} y_2(x) = y_2(x) \quad (1.3)$$

```
> Out1:=NLDE:-arithmeticDalg([ADE1,ADE2],[y[1](x),y[2](x)],z=y[1]+y
  [2])
```

$$Out1 := \frac{d^3}{dx^3} z(x) - \frac{d^2}{dx^2} z(x) + \frac{d}{dx} z(x) - z(x) = 0 \quad (1.4)$$

```
> Out2:=NLDE:-arithmeticDalg([ADE1,ADE2],[y[1](x),y[2](x)],z=y[1]+y
  [2],lho=false):
> factor(Out2)
```

$$\left(\frac{d}{dx} z(x) - z(x) + 1\right)\left(\frac{d}{dx} z(x) - z(x) - 1\right)\left(\left(\frac{d^2}{dx^2} z(x)\right)^2 - 2\left(\frac{d}{dx} z(x)\right)\left(\frac{d^2}{dx^2} z(x)\right) + 2\left(\frac{d}{dx} z(x)\right)^2 - 2 z(x)\left(\frac{d}{dx} z(x)\right) + z(x)^2 - 2\right) = 0 \quad (1.5)$$

Example 5.

```
> ADE3:=diff(y[3](x),x)^3+diff(y[3](x),x)^2+3=0
```

$$ADE3 := \left(\frac{d}{dx} y_3(x)\right)^3 + \left(\frac{d}{dx} y_3(x)\right)^2 + 3 = 0 \quad \textbf{(2.1)}$$

```
> Out3:=NLDE:-arithmeticDalg([ADE1,ADE2,ADE3],[y[1](x),y[2](x),y[3]
  (x)],z=y[1]*y[3]/y[2])
```

$$Out3 := 12\,z(x)^2 + 32\,z(x)\left(\frac{d}{dx} z(x)\right) + 20\left(\frac{d^2}{dx^2} z(x)\right) z(x) + 6\left(\frac{d^3}{dx^3} z(x)\right) z(x) + 24\left(\frac{d}{dx} z(x)\right)^2 + 30\left(\frac{d}{dx} z(x)\right)\left(\frac{d^2}{dx^2} z(x)\right) + 10\left(\frac{d^3}{dx^3} z(x)\right)\left(\frac{d}{dx} z(x)\right) + 9\left(\frac{d^2}{dx^2} z(x)\right)^2 + 6\left(\frac{d^3}{dx^3} z(x)\right)\left(\frac{d^2}{dx^2} z(x)\right) + \left(\frac{d^3}{dx^3} z(x)\right)^2 = 0 \quad \textbf{(2.2)}$$

```
> Out4:=NLDE:-arithmeticDalg([ADE1,ADE2,ADE3],[y[1](x),y[2](x),y[3]
  (x)],z=y[1]*y[3]/y[2],lho=false):
> factor(Out4)
```

$$\left(\frac{d^2}{dx^2} z(x) + 2\frac{d}{dx} z(x) + z(x)\right)\left(\frac{d^2}{dx^2} z(x) + 2\frac{d}{dx} z(x) + 2\,z(x)\right)^2 \left(12\,z(x)^2 + 32\,z(x)\left(\frac{d}{dx} z(x)\right) + 20\left(\frac{d^2}{dx^2} z(x)\right) z(x) + 6\left(\frac{d^3}{dx^3} z(x)\right) z(x) + 24\left(\frac{d}{dx} z(x)\right)^2 + 30\left(\frac{d}{dx} z(x)\right)\left(\frac{d^2}{dx^2} z(x)\right) + 10\left(\frac{d^3}{dx^3} z(x)\right)\left(\frac{d}{dx} z(x)\right) + 9\left(\frac{d^2}{dx^2} z(x)\right)^2 + 6\left(\frac{d^3}{dx^3} z(x)\right)\left(\frac{d^2}{dx^2} z(x)\right) + \left(\frac{d^3}{dx^3} z(x)\right)^2\right) = 0 \quad \textbf{(2.3)}$$

Example 8.

```
> with(NLDE:-MultiDalg): #to load the subpackage
> ADE1:=diff(U(x,y,z),x)=diff(U(x,y,z),y):
> ADE2:=diff(V(x,y,z),y)=diff(V(x,y,z),z):
> ADE3:=diff(W(x,y,z),z)=diff(W(x,y,z),x):
> arithmeticMDalg([ADE1,ADE2,ADE3],[U(x,y,z),V(x,y,z),W(x,y,z)],T=
  U+V+W)
```

$$\frac{\partial^3}{\partial x^2 \partial y} T(x,y,z) - \frac{\partial^3}{\partial x \partial y^2} T(x,y,z) - \frac{\partial^3}{\partial x^2 \partial z} T(x,y,z) + \frac{\partial^3}{\partial y^2 \partial z} T(x,y,z) + \frac{\partial^3}{\partial x \partial z^2} T(x,y,z) - \frac{\partial^3}{\partial y \partial z^2} T(x,y,z) = 0 \quad \textbf{(3.1)}$$

Example 9.

```
> ADE1:=diff(y[1](x[1],x[2]),x[1],x[2])*x[2]+diff(y[1](x[1],x[2]),x[2])=0:
> ADE2:=diff(y[2](x[1],x[2]),x[1])*x[1]-diff(y[2](x[1],x[2]),x[1],x[1])=0:
> arithmeticMDalg([ADE1,ADE2],[y[1](x[1],x[2]),y[2](x[1],x[2])],z=y[1]+y[2])
```

$$\left(x_1^2 x_2 + x_1 + x_2\right)\left(\frac{\partial^2}{\partial x_1 \partial x_2} z(x_1, x_2)\right) + \left(x_1^2 x_2^2 + x_2^2 - 1\right)\left(\frac{\partial^3}{\partial x_1^2 \partial x_2} z(x_1, x_2)\right) + \left(-x_1 x_2^2 - x_2\right)\left(\frac{\partial^4}{\partial x_1^3 \partial x_2} z(x_1, x_2)\right) = 0 \quad \textbf{(4.1)}$$

```
> arithmeticMDalg([ADE1,ADE2],[y[1](x[1],x[2]),y[2](x[1],x[2])],z=y[1]+y[2],maxord=[4,1])
```

$$\left(x_1^3 x_2 + x_1^2 + 3\, x_1 x_2 + 1\right)\left(\frac{\partial^3}{\partial x_1^2 \partial x_2} z(x_1, x_2)\right) + \left(\frac{\partial^4}{\partial x_1^3 \partial x_2} z(x_1, x_2)\right)\left(x_1^3 x_2^2 + 3\, x_1 x_2^2 - x_1\right) + \left(\frac{\partial^5}{\partial x_1^4 \partial x_2} z(x_1, x_2)\right)\left(-x_1^2 x_2^2 - x_1 x_2 - x_2^2\right) = 0 \quad \textbf{(4.2)}$$

Example 10.

```
> ADEKV:=-c*diff(v(x),x)+diff(v(x),x,x,x)+6*v(x)*diff(v(x),x)=0:
> NLDE:-unaryDalg(ADEKV,v(x),w=C[1]*v+C[2])
```

$$6\, w(x)\left(\frac{d}{dx} w(x)\right) + \left(-c\, C_1 - 6\, C_2\right)\left(\frac{d}{dx} w(x)\right) + C_1\left(\frac{d^3}{dx^3} w(x)\right) = 0 \quad \textbf{(5.1)}$$

```
> NLDE:-unaryDalg(ADEKV,v(x),w=-v+c/6)
```

$$6\, w(x)\left(\frac{d}{dx} w(x)\right) - \frac{d^3}{dx^3} w(x) = 0 \quad \textbf{(5.2)}$$

```
> NLDE:-unaryDalg(ADEKV,v(x),w=(-6*v+c)/12)
```

$$12\, w(x)\left(\frac{d}{dx} w(x)\right) - \frac{d^3}{dx^3} w(x) = 0 \quad \textbf{(5.3)}$$

Example 11.

```
> ADEwp:=diff(p(x),x)^2=4*p(x)^3-g2*p(x)-g3:
> ADE1:=NLDE:-unaryDalg(ADEwp,p(x),v=-2*p+c/6)
```

$$ADE1 := 216\, v(x)^3 - 108\, v(x)^2 c + \left(18\, c^2 - 216\, g2\right) v(x) + 108\left(\frac{d}{dx} v(x)\right)^2 - c^3 + 36\, c\, g2 + 432\, g3 = 0 \quad \textbf{(6.1)}$$

```
> factor(diff(ADE1,x))
```

$$18\left(\frac{d}{dx}v(x)\right)\left(36 v(x)^2 - 12 v(x) c + c^2 + 12 \frac{d^2}{dx^2} v(x) - 12 g2\right) = 0 \quad \textbf{(6.2)}$$

Example 12.

```
> ADE1:=-a*y[1](x[1]) + diff(y[1](x[1]), x[1]) = 0:
> ADE2:=-b*y[2](x[2]) + diff(y[2](x[2]), x[2]) = 0:
> arithmeticMDalg([ADE1,ADE2],[y[1](x[1]),y[2](x[2])],z=1/(1+y[1]*y[2]),[x[1],x[2]])
```

$$z(x_1, x_2)^2 b - b z(x_1, x_2) - \frac{\partial}{\partial x_2} z(x_1, x_2) = 0 \quad \textbf{(7.1)}$$

```
> pdsolve(%)
```

$$z(x_1, x_2) = \frac{1}{1 + e^{b x_2} \_F1(x_1)} \quad \textbf{(7.2)}$$

Example 13.

```
> arithmeticMDalg([ADE1,ADE2],[y[1](x[1]),y[2](x[2])],z=y[1]/(1+y[1]*y[2]),[x[1],x[2]])
```

$$b a z(x_1, x_2) + \left(\frac{\partial}{\partial x_2} z(x_1, x_2)\right) a - \left(\frac{\partial}{\partial x_1} z(x_1, x_2)\right) b = 0 \quad \textbf{(8.1)}$$

```
> pdsolve(%)
```

$$z(x_1, x_2) = \_F1\left(\frac{x_1 a + b x_2}{b}\right) e^{x_1 a} \quad \textbf{(8.2)}$$

Bonus Example: PDE fulfilled by the square of the probability density function of the univariate normal distribution, seen as a bivariate function in the mean and the indeterminate variable.

For the use of the FPS software below, please download the most recent version at FPS web page. One can also use DEtools:-FindODE to compute **(9.2)** and **(9.3)** in the univariate way.

```
> f:=exp(-((x-mu)/sigma)^2/2)/(sigma*sqrt(2*Pi))
```

(9.1)

$$f := \frac{e^{-\frac{(x-\mu)^2}{2\sigma^2}}\sqrt{2}}{2\,\sigma\sqrt{\pi}} \tag{9.1}$$

```
> PDE1:=FPS:-HolonomicPDE(f,u(x,mu),partialwrt=x)
```

$$PDE1 := \sigma^2\left(\frac{\partial}{\partial x}\,u(x,\mu)\right) - (-x+\mu)\,u(x,\mu) = 0 \tag{9.2}$$

```
> PDE2:=FPS:-HolonomicPDE(f,v(x,mu),partialwrt=mu)
```

$$PDE2 := \sigma^2\left(\frac{\partial}{\partial \mu}\,v(x,\mu)\right) - (x-\mu)\,v(x,\mu) = 0 \tag{9.3}$$

```
> arithmeticMDalg([PDE1,PDE2],[u(x,mu),v(x,mu)],z=u*v)
```

$$-\sigma^2 z(x,\mu)\left(\frac{\partial^2}{\partial\mu\,\partial x}\,z(x,\mu)\right) + \left(\frac{\partial}{\partial \mu}\,z(x,\mu)\right)\left(\frac{\partial}{\partial x}\,z(x,\mu)\right)\sigma^2 + 2\,z(x,\mu)^2 = 0 \tag{9.4}$$